

# The Hahn-Banach Theorem for Real Vector Spaces

Gertrud Bauer

September 11, 2023

## Abstract

The Hahn-Banach Theorem is one of the most fundamental results in functional analysis. We present a fully formal proof of two versions of the theorem, one for general linear spaces and another for normed spaces. This development is based on simply-typed classical set-theory, as provided by Isabelle/HOL.

## Contents


<b>1</b>	<b>Preface</b>	<b>3</b>
<b>I</b>	<b>Basic Notions</b>	<b>5</b>
<b>2</b>	<b>Bounds</b>	<b>5</b>
<b>3</b>	<b>Vector spaces</b>	<b>5</b>
3.1	Signature . . . . .	5
3.2	Vector space laws . . . . .	6
<b>4</b>	<b>Subspaces</b>	<b>9</b>
4.1	Definition . . . . .	9
4.2	Linear closure . . . . .	10
4.3	Sum of two vectorspaces . . . . .	11
4.4	Direct sums . . . . .	12
<b>5</b>	<b>Normed vector spaces</b>	<b>13</b>
5.1	Quasinorms . . . . .	13
5.2	Norms . . . . .	13
5.3	Normed vector spaces . . . . .	13
<b>6</b>	<b>Linearforms</b>	<b>14</b>
<b>7</b>	<b>An order on functions</b>	<b>14</b>
7.1	The graph of a function . . . . .	14
7.2	Functions ordered by domain extension . . . . .	15
7.3	Domain and function of a graph . . . . .	15
7.4	Norm-preserving extensions of a function . . . . .	15

<b>8</b>	<b>The norm of a function</b>	<b>16</b>
8.1	Continuous linear forms . . . . .	16
8.2	The norm of a linear form . . . . .	17
<b>9</b>	<b>Zorn's Lemma</b>	<b>18</b>
<b>II</b>	<b>Lemmas for the Proof</b>	<b>20</b>
<b>10</b>	<b>The supremum wrt. the function order</b>	<b>20</b>
<b>11</b>	<b>Extending non-maximal functions</b>	<b>22</b>
<b>III</b>	<b>The Main Proof</b>	<b>24</b>
<b>12</b>	<b>The Hahn-Banach Theorem</b>	<b>24</b>
12.1	The Hahn-Banach Theorem for vector spaces . . . . .	24
12.2	Alternative formulation . . . . .	24
12.3	The Hahn-Banach Theorem for normed spaces . . . . .	25

# 1 Preface

This is a fully formal proof of the Hahn-Banach Theorem. It closely follows the informal presentation given in Heuser's textbook [1, § 36]. Another formal proof of the same theorem has been done in Mizar [3]. A general overview of the relevance and history of the Hahn-Banach Theorem is given by Narici and Beckenstein [2].

The document is structured as follows. The first part contains definitions of basic notions of linear algebra: vector spaces, subspaces, normed spaces, continuous linear-forms, norm of functions and an order on functions by domain extension. The second part contains some lemmas about the supremum (w.r.t. the function order) and extension of non-maximal functions. With these preliminaries, the main proof of the theorem (in its two versions) is conducted in the third part. The dependencies of individual theories are as follows.



# Part I

## Basic Notions

### 2 Bounds

```

theory Bounds
imports Main HOL-Analysis.Continuum-Not-Denumerable
begin

locale lub =
  fixes A and x
  assumes least [intro?]: ( $\bigwedge a. a \in A \implies a \leq b$ )  $\implies x \leq b$ 
  and upper [intro?]:  $a \in A \implies a \leq x$ 

lemmas [elim?] = lub.least lub.upper

definition the-lub :: 'a::order set  $\Rightarrow$  'a ( $\lfloor$  - [90] 90)
  where the-lub A = The (lub A)

lemma the-lub-equality [elim?]:
  assumes lub A x
  shows  $\lfloor A = (x::'a::order)$ 
  <proof>

lemma the-lubI-ex:
  assumes ex:  $\exists x. \text{lub } A \ x$ 
  shows lub A ( $\lfloor A$ )
  <proof>

lemma real-complete:  $\exists a::\text{real}. a \in A \implies \exists y. \forall a \in A. a \leq y \implies \exists x. \text{lub } A \ x$ 
  <proof>

end

```

### 3 Vector spaces

```

theory Vector-Space
imports Complex-Main Bounds
begin

```

#### 3.1 Signature

For the definition of real vector spaces a type 'a of the sort {plus, minus, zero} is considered, on which a real scalar multiplication  $\cdot$  is declared.

```

consts
  prod :: real  $\Rightarrow$  'a::{plus,minus,zero}  $\Rightarrow$  'a (infixr  $\cdot$  70)

```

### 3.2 Vector space laws

A *vector space* is a non-empty set  $V$  of elements from  $'a$  with the following vector space laws: The set  $V$  is closed under addition and scalar multiplication, addition is associative and commutative;  $-x$  is the inverse of  $x$  wrt. addition and  $0$  is the neutral element of addition. Addition and multiplication are distributive; scalar multiplication is associative and the real number  $1$  is the neutral element of scalar multiplication.

```

locale vectorspace =
  fixes V
  assumes non-empty [iff, intro?]: V ≠ {}
  and add-closed [iff]: x ∈ V ⇒ y ∈ V ⇒ x + y ∈ V
  and mult-closed [iff]: x ∈ V ⇒ a · x ∈ V
  and add-assoc: x ∈ V ⇒ y ∈ V ⇒ z ∈ V ⇒ (x + y) + z = x + (y + z)
  and add-commute: x ∈ V ⇒ y ∈ V ⇒ x + y = y + x
  and diff-self [simp]: x ∈ V ⇒ x - x = 0
  and add-zero-left [simp]: x ∈ V ⇒ 0 + x = x
  and add-mult-distrib1: x ∈ V ⇒ y ∈ V ⇒ a · (x + y) = a · x + a · y
  and add-mult-distrib2: x ∈ V ⇒ (a + b) · x = a · x + b · x
  and mult-assoc: x ∈ V ⇒ (a * b) · x = a · (b · x)
  and mult-1 [simp]: x ∈ V ⇒ 1 · x = x
  and negate-eq1: x ∈ V ⇒ - x = (- 1) · x
  and diff-eq1: x ∈ V ⇒ y ∈ V ⇒ x - y = x + - y
begin

```

**lemma** negate-eq2:  $x \in V \Rightarrow (- 1) \cdot x = - x$   
 ⟨proof⟩

**lemma** negate-eq2a:  $x \in V \Rightarrow -1 \cdot x = - x$   
 ⟨proof⟩

**lemma** diff-eq2:  $x \in V \Rightarrow y \in V \Rightarrow x + - y = x - y$   
 ⟨proof⟩

**lemma** diff-closed [iff]:  $x \in V \Rightarrow y \in V \Rightarrow x - y \in V$   
 ⟨proof⟩

**lemma** neg-closed [iff]:  $x \in V \Rightarrow - x \in V$   
 ⟨proof⟩

**lemma** add-left-commute:  
 $x \in V \Rightarrow y \in V \Rightarrow z \in V \Rightarrow x + (y + z) = y + (x + z)$   
 ⟨proof⟩

**lemmas** add-ac = add-assoc add-commute add-left-commute

The existence of the zero element of a vector space follows from the non-emptiness of carrier set.

**lemma** zero [iff]:  $0 \in V$   
 ⟨proof⟩

**lemma** add-zero-right [simp]:  $x \in V \Rightarrow x + 0 = x$   
 ⟨proof⟩

**lemma** *mult-assoc2*:  $x \in V \implies a \cdot b \cdot x = (a * b) \cdot x$   
 ⟨proof⟩

**lemma** *diff-mult-distrib1*:  $x \in V \implies y \in V \implies a \cdot (x - y) = a \cdot x - a \cdot y$   
 ⟨proof⟩

**lemma** *diff-mult-distrib2*:  $x \in V \implies (a - b) \cdot x = a \cdot x - (b \cdot x)$   
 ⟨proof⟩

**lemmas** *distrib =*  
*add-mult-distrib1 add-mult-distrib2*  
*diff-mult-distrib1 diff-mult-distrib2*

Further derived laws:

**lemma** *mult-zero-left [simp]*:  $x \in V \implies 0 \cdot x = 0$   
 ⟨proof⟩

**lemma** *mult-zero-right [simp]*:  $a \cdot 0 = (0::'a)$   
 ⟨proof⟩

**lemma** *minus-mult-cancel [simp]*:  $x \in V \implies (- a) \cdot - x = a \cdot x$   
 ⟨proof⟩

**lemma** *add-minus-left-eq-diff*:  $x \in V \implies y \in V \implies - x + y = y - x$   
 ⟨proof⟩

**lemma** *add-minus [simp]*:  $x \in V \implies x + - x = 0$   
 ⟨proof⟩

**lemma** *add-minus-left [simp]*:  $x \in V \implies - x + x = 0$   
 ⟨proof⟩

**lemma** *minus-minus [simp]*:  $x \in V \implies - (- x) = x$   
 ⟨proof⟩

**lemma** *minus-zero [simp]*:  $- (0::'a) = 0$   
 ⟨proof⟩

**lemma** *minus-zero-iff [simp]*:  
**assumes**  $x: x \in V$   
**shows**  $(- x = 0) = (x = 0)$   
 ⟨proof⟩

**lemma** *add-minus-cancel [simp]*:  $x \in V \implies y \in V \implies x + (- x + y) = y$   
 ⟨proof⟩

**lemma** *minus-add-cancel [simp]*:  $x \in V \implies y \in V \implies - x + (x + y) = y$   
 ⟨proof⟩

**lemma** *minus-add-distrib [simp]*:  $x \in V \implies y \in V \implies - (x + y) = - x + - y$   
 ⟨proof⟩

**lemma** *diff-zero [simp]*:  $x \in V \implies x - 0 = x$

*<proof>*

**lemma** *diff-zero-right* [*simp*]:  $x \in V \implies 0 - x = -x$

*<proof>*

**lemma** *add-left-cancel*:

**assumes**  $x: x \in V$  **and**  $y: y \in V$  **and**  $z: z \in V$

**shows**  $(x + y = x + z) = (y = z)$

*<proof>*

**lemma** *add-right-cancel*:

$x \in V \implies y \in V \implies z \in V \implies (y + x = z + x) = (y = z)$

*<proof>*

**lemma** *add-assoc-cong*:

$x \in V \implies y \in V \implies x' \in V \implies y' \in V \implies z \in V$

$\implies x + y = x' + y' \implies x + (y + z) = x' + (y' + z)$

*<proof>*

**lemma** *mult-left-commute*:  $x \in V \implies a \cdot b \cdot x = b \cdot a \cdot x$

*<proof>*

**lemma** *mult-zero-uniq*:

**assumes**  $x: x \in V$   $x \neq 0$  **and**  $ax: a \cdot x = 0$

**shows**  $a = 0$

*<proof>*

**lemma** *mult-left-cancel*:

**assumes**  $x: x \in V$  **and**  $y: y \in V$  **and**  $a: a \neq 0$

**shows**  $(a \cdot x = a \cdot y) = (x = y)$

*<proof>*

**lemma** *mult-right-cancel*:

**assumes**  $x: x \in V$  **and**  $neg: x \neq 0$

**shows**  $(a \cdot x = b \cdot x) = (a = b)$

*<proof>*

**lemma** *eq-diff-eq*:

**assumes**  $x: x \in V$  **and**  $y: y \in V$  **and**  $z: z \in V$

**shows**  $(x = z - y) = (x + y = z)$

*<proof>*

**lemma** *add-minus-eq-minus*:

**assumes**  $x: x \in V$  **and**  $y: y \in V$  **and**  $xy: x + y = 0$

**shows**  $x = -y$

*<proof>*

**lemma** *add-minus-eq*:

**assumes**  $x: x \in V$  **and**  $y: y \in V$  **and**  $xy: x - y = 0$

**shows**  $x = y$

*<proof>*

**lemma** *add-diff-swap*:

**assumes**  $vs: a \in V$   $b \in V$   $c \in V$   $d \in V$



**and**  $eq: a + b = c + d$   
**shows**  $a - c = d - b$   
 $\langle proof \rangle$

**lemma** *vs-add-cancel-21*:  
**assumes**  $vs: x \in V \ y \in V \ z \in V \ u \in V$   
**shows**  $(x + (y + z) = y + u) = (x + z = u)$   
 $\langle proof \rangle$

**lemma** *add-cancel-end*:  
**assumes**  $vs: x \in V \ y \in V \ z \in V$   
**shows**  $(x + (y + z) = y) = (x = -z)$   
 $\langle proof \rangle$

**end**

**end**

## 4 Subspaces

**theory** *Subspace*  
**imports** *Vector-Space HOL-Library.Set-Algebras*  
**begin**

### 4.1 Definition

A non-empty subset  $U$  of a vector space  $V$  is a *subspace* of  $V$ , iff  $U$  is closed under addition and scalar multiplication.

**locale** *subspace* =  
**fixes**  $U :: 'a::\{minus, plus, zero, uminus\}$  *set* **and**  $V$   
**assumes** *non-empty* [*iff, intro*]:  $U \neq \{\}$   
**and** *subset* [*iff*]:  $U \subseteq V$   
**and** *add-closed* [*iff*]:  $x \in U \implies y \in U \implies x + y \in U$   
**and** *mult-closed* [*iff*]:  $x \in U \implies a \cdot x \in U$

**notation** (*symbols*)  
*subspace* (**infix**  $\trianglelefteq$  50)

**declare** *vectorspace.intro* [*intro?*] *subspace.intro* [*intro?*]

**lemma** *subspace-subset* [*elim*]:  $U \trianglelefteq V \implies U \subseteq V$   
 $\langle proof \rangle$

**lemma** (**in** *subspace*) *subsetD* [*iff*]:  $x \in U \implies x \in V$   
 $\langle proof \rangle$

**lemma** *subspaceD* [*elim*]:  $U \trianglelefteq V \implies x \in U \implies x \in V$   
 $\langle proof \rangle$

**lemma** *rev-subspaceD* [*elim?*]:  $x \in U \implies U \trianglelefteq V \implies x \in V$   
 $\langle proof \rangle$

**lemma** (**in** *subspace*) *diff-closed* [*iff*]:

```

assumes vectorspace  $V$ 
assumes  $x: x \in U$  and  $y: y \in U$ 
shows  $x - y \in U$ 
<proof>

```

Similar as for linear spaces, the existence of the zero element in every subspace follows from the non-emptiness of the carrier set and by vector space laws.

```

lemma (in subspace) zero [intro]:
  assumes vectorspace  $V$ 
  shows  $0 \in U$ 
<proof>

```

```

lemma (in subspace) neg-closed [iff]:
  assumes vectorspace  $V$ 
  assumes  $x: x \in U$ 
  shows  $-x \in U$ 
<proof>

```

Further derived laws: every subspace is a vector space.

```

lemma (in subspace) vectorspace [iff]:
  assumes vectorspace  $V$ 
  shows vectorspace  $U$ 
<proof>

```

The subspace relation is reflexive.

```

lemma (in vectorspace) subspace-refl [intro]:  $V \trianglelefteq V$ 
<proof>

```

The subspace relation is transitive.

```

lemma (in vectorspace) subspace-trans [trans]:
   $U \trianglelefteq V \implies V \trianglelefteq W \implies U \trianglelefteq W$ 
<proof>

```

## 4.2 Linear closure

The *linear closure* of a vector  $x$  is the set of all scalar multiples of  $x$ .

```

definition lin :: ('a::{minus,plus,zero})  $\Rightarrow$  'a set
  where  $lin\ x = \{a \cdot x \mid a. True\}$ 

```

```

lemma linI [intro]:  $y = a \cdot x \implies y \in lin\ x$ 
<proof>

```

```

lemma linI' [iff]:  $a \cdot x \in lin\ x$ 
<proof>

```

```

lemma linE [elim]:
  assumes  $x \in lin\ v$ 
  obtains  $a :: real$  where  $x = a \cdot v$ 
<proof>

```

Every vector is contained in its linear closure.

**lemma** (in *vectorpace*) *x-lin-x* [iff]:  $x \in V \implies x \in \text{lin } x$   
 ⟨proof⟩

**lemma** (in *vectorpace*) *0-lin-x* [iff]:  $x \in V \implies 0 \in \text{lin } x$   
 ⟨proof⟩

Any linear closure is a subspace.

**lemma** (in *vectorpace*) *lin-subspace* [intro]:  
 assumes  $x: x \in V$   
 shows  $\text{lin } x \trianglelefteq V$   
 ⟨proof⟩

Any linear closure is a vector space.

**lemma** (in *vectorpace*) *lin-vectorpace* [intro]:  
 assumes  $x \in V$   
 shows *vectorpace* ( $\text{lin } x$ )  
 ⟨proof⟩

### 4.3 Sum of two vectorspaces

The *sum* of two vectorspaces  $U$  and  $V$  is the set of all sums of elements from  $U$  and  $V$ .

**lemma** *sum-def*:  $U + V = \{u + v \mid u \in U \wedge v \in V\}$   
 ⟨proof⟩

**lemma** *sumE* [elim]:  
 $x \in U + V \implies (\bigwedge u \in U, v \in V. x = u + v \implies C) \implies C$   
 ⟨proof⟩

**lemma** *sumI* [intro]:  
 $u \in U \implies v \in V \implies x = u + v \implies x \in U + V$   
 ⟨proof⟩

**lemma** *sumI'* [intro]:  
 $u \in U \implies v \in V \implies u + v \in U + V$   
 ⟨proof⟩

$U$  is a subspace of  $U + V$ .

**lemma** *subspace-sumI* [iff]:  
 assumes *vectorpace*  $U$  *vectorpace*  $V$   
 shows  $U \trianglelefteq U + V$   
 ⟨proof⟩

The sum of two subspaces is again a subspace.

**lemma** *sum-subspace* [intro?]:  
 assumes *subspace*  $U$  *vectorpace*  $E$  *subspace*  $V$   $E$   
 shows  $U + V \trianglelefteq E$   
 ⟨proof⟩

The sum of two subspaces is a vectorpace.

**lemma** *sum-vs* [intro?]:  
 $U \trianglelefteq E \implies V \trianglelefteq E \implies \text{vectorpace } E \implies \text{vectorpace } (U + V)$   
 ⟨proof⟩

#### 4.4 Direct sums

The sum of  $U$  and  $V$  is called *direct*, iff the zero element is the only common element of  $U$  and  $V$ . For every element  $x$  of the direct sum of  $U$  and  $V$  the decomposition in  $x = u + v$  with  $u \in U$  and  $v \in V$  is unique.

**lemma** *decomp*:

**assumes** *vectorspace*  $E$  *subspace*  $U$  *subspace*  $V$   $E$

**assumes** *direct*:  $U \cap V = \{0\}$

**and**  $u1$ :  $u1 \in U$  **and**  $u2$ :  $u2 \in U$

**and**  $v1$ :  $v1 \in V$  **and**  $v2$ :  $v2 \in V$

**and** *sum*:  $u1 + v1 = u2 + v2$

**shows**  $u1 = u2 \wedge v1 = v2$

*<proof>*

An application of the previous lemma will be used in the proof of the Hahn-Banach Theorem (see page ??): for any element  $y + a \cdot x_0$  of the direct sum of a vectorspace  $H$  and the linear closure of  $x_0$  the components  $y \in H$  and  $a$  are uniquely determined.

**lemma** *decomp-H'*:

**assumes** *vectorspace*  $E$  *subspace*  $H$   $E$

**assumes**  $y1$ :  $y1 \in H$  **and**  $y2$ :  $y2 \in H$

**and**  $x'$ :  $x' \notin H$   $x' \in E$   $x' \neq 0$

**and** *eq*:  $y1 + a1 \cdot x' = y2 + a2 \cdot x'$

**shows**  $y1 = y2 \wedge a1 = a2$

*<proof>*

Since for any element  $y + a \cdot x'$  of the direct sum of a vectorspace  $H$  and the linear closure of  $x'$  the components  $y \in H$  and  $a$  are unique, it follows from  $y \in H$  that  $a = 0$ .

**lemma** *decomp-H'-H*:

**assumes** *vectorspace*  $E$  *subspace*  $H$   $E$

**assumes**  $t$ :  $t \in H$

**and**  $x'$ :  $x' \notin H$   $x' \in E$   $x' \neq 0$

**shows** (*SOME*  $(y, a)$ .  $t = y + a \cdot x' \wedge y \in H$ ) =  $(t, 0)$

*<proof>*

The components  $y \in H$  and  $a$  in  $y + a \cdot x'$  are unique, so the function  $h'$  defined by  $h'(y + a \cdot x') = h y + a \cdot \xi$  is definite.

**lemma** *h'-definite*:

**fixes**  $H$

**assumes** *h'-def*:

$\bigwedge x$ .  $h' x =$

(*let*  $(y, a) = \text{SOME } (y, a)$ .  $(x = y + a \cdot x' \wedge y \in H)$

*in*  $(h y) + a * xi$ )

**and**  $x$ :  $x = y + a \cdot x'$

**assumes** *vectorspace*  $E$  *subspace*  $H$   $E$

**assumes**  $y$ :  $y \in H$

**and**  $x'$ :  $x' \notin H$   $x' \in E$   $x' \neq 0$

**shows**  $h' x = h y + a * xi$

*<proof>*

**end**

## 5 Normed vector spaces

```
theory Normed-Space
imports Subspace
begin
```

### 5.1 Quasinorms

A *seminorm*  $\|\cdot\|$  is a function on a real vector space into the reals that has the following properties: it is positive definite, absolute homogeneous and subadditive.

```
locale seminorm =
  fixes V :: 'a::{minus, plus, zero, uminus} set
  fixes norm :: 'a  $\Rightarrow$  real ( $\|\cdot\|$ )
  assumes ge-zero [iff?]:  $x \in V \Rightarrow 0 \leq \|x\|$ 
    and abs-homogenous [iff?]:  $x \in V \Rightarrow \|a \cdot x\| = |a| * \|x\|$ 
    and subadditive [iff?]:  $x \in V \Rightarrow y \in V \Rightarrow \|x + y\| \leq \|x\| + \|y\|$ 
```

```
declare seminorm.intro [intro?]
```

```
lemma (in seminorm) diff-subadditive:
  assumes vectorspace V
  shows  $x \in V \Rightarrow y \in V \Rightarrow \|x - y\| \leq \|x\| + \|y\|$ 
<proof>
```

```
lemma (in seminorm) minus:
  assumes vectorspace V
  shows  $x \in V \Rightarrow \|- x\| = \|x\|$ 
<proof>
```

### 5.2 Norms

A *norm*  $\|\cdot\|$  is a seminorm that maps only the  $0$  vector to  $0$ .

```
locale norm = seminorm +
  assumes zero-iff [iff]:  $x \in V \Rightarrow (\|x\| = 0) = (x = 0)$ 
```

### 5.3 Normed vector spaces

A vector space together with a norm is called a *normed space*.

```
locale normed-vectorspace = vectorspace + norm
```

```
declare normed-vectorspace.intro [intro?]
```

```
lemma (in normed-vectorspace) gt-zero [intro?]:
  assumes  $x: x \in V$  and neq:  $x \neq 0$ 
  shows  $0 < \|x\|$ 
<proof>
```

Any subspace of a normed vector space is again a normed vectorspace.

```
lemma subspace-normed-vs [intro?]:
  fixes F E norm
  assumes subspace F E normed-vectorspace E norm
```

```

shows normed-vectorspace F norm
<proof>

```

```

end

```

## 6 Linearforms

```

theory Linearform
imports Vector-Space
begin

```

A *linear form* is a function on a vector space into the reals that is additive and multiplicative.

```

locale linearform =
  fixes  $V :: 'a::\{minus, plus, zero, uminus\}$  set and  $f$ 
  assumes add [iff]:  $x \in V \implies y \in V \implies f(x + y) = f x + f y$ 
  and mult [iff]:  $x \in V \implies f(a \cdot x) = a * f x$ 

```

```

declare linearform.intro [intro?]

```

```

lemma (in linearform) neg [iff]:
  assumes vectorspace V
  shows  $x \in V \implies f(-x) = -f x$ 
<proof>

```

```

lemma (in linearform) diff [iff]:
  assumes vectorspace V
  shows  $x \in V \implies y \in V \implies f(x - y) = f x - f y$ 
<proof>

```

Every linear form yields 0 for the 0 vector.

```

lemma (in linearform) zero [iff]:
  assumes vectorspace V
  shows  $f 0 = 0$ 
<proof>

```

```

end

```

## 7 An order on functions

```

theory Function-Order
imports Subspace Linearform
begin

```

### 7.1 The graph of a function

We define the *graph* of a (real) function  $f$  with domain  $F$  as the set

$$\{(x, f x). x \in F\}$$

So we are modeling partial functions by specifying the domain and the mapping function. We use the term “function” also for its graph.

**type-synonym**  $'a \text{ graph} = ('a \times \text{real}) \text{ set}$

**definition**  $\text{graph} :: 'a \text{ set} \Rightarrow ('a \Rightarrow \text{real}) \Rightarrow 'a \text{ graph}$   
**where**  $\text{graph } F f = \{(x, f x) \mid x. x \in F\}$

**lemma**  $\text{graphI}$  [*intro*]:  $x \in F \Longrightarrow (x, f x) \in \text{graph } F f$   
*<proof>*

**lemma**  $\text{graphI2}$  [*intro?*]:  $x \in F \Longrightarrow \exists t \in \text{graph } F f. t = (x, f x)$   
*<proof>*

**lemma**  $\text{graphE}$  [*elim?*]:  
**assumes**  $(x, y) \in \text{graph } F f$   
**obtains**  $x \in F$  **and**  $y = f x$   
*<proof>*

## 7.2 Functions ordered by domain extension

A function  $h'$  is an extension of  $h$ , iff the graph of  $h$  is a subset of the graph of  $h'$ .

**lemma**  $\text{graph-extI}$ :  
 $(\bigwedge x. x \in H \Longrightarrow h x = h' x) \Longrightarrow H \subseteq H'$   
 $\Longrightarrow \text{graph } H h \subseteq \text{graph } H' h'$   
*<proof>*

**lemma**  $\text{graph-extD1}$  [*dest?*]:  $\text{graph } H h \subseteq \text{graph } H' h' \Longrightarrow x \in H \Longrightarrow h x = h' x$   
*<proof>*

**lemma**  $\text{graph-extD2}$  [*dest?*]:  $\text{graph } H h \subseteq \text{graph } H' h' \Longrightarrow H \subseteq H'$   
*<proof>*

## 7.3 Domain and function of a graph

The inverse functions to  $\text{graph}$  are *domain* and *funct*.

**definition**  $\text{domain} :: 'a \text{ graph} \Rightarrow 'a \text{ set}$   
**where**  $\text{domain } g = \{x. \exists y. (x, y) \in g\}$

**definition**  $\text{funct} :: 'a \text{ graph} \Rightarrow ('a \Rightarrow \text{real})$   
**where**  $\text{funct } g = (\lambda x. (\text{SOME } y. (x, y) \in g))$

The following lemma states that  $g$  is the graph of a function if the relation induced by  $g$  is unique.

**lemma**  $\text{graph-domain-funct}$ :  
**assumes**  $\text{uniq}: \bigwedge x y z. (x, y) \in g \Longrightarrow (x, z) \in g \Longrightarrow z = y$   
**shows**  $\text{graph } (\text{domain } g) (\text{funct } g) = g$   
*<proof>*

## 7.4 Norm-preserving extensions of a function

Given a linear form  $f$  on the space  $F$  and a seminorm  $p$  on  $E$ . The set of all linear extensions of  $f$ , to superspaces  $H$  of  $F$ , which are bounded by  $p$ , is defined as follows.

**definition**

*norm-pres-extensions* ::  
 'a::{plus,minus,uminus,zero} set  $\Rightarrow$  ('a  $\Rightarrow$  real)  $\Rightarrow$  'a set  $\Rightarrow$  ('a  $\Rightarrow$  real)  
 $\Rightarrow$  'a graph set

**where**

*norm-pres-extensions* E p F f  
 = {g.  $\exists H h. g = \text{graph } H h$   
 $\wedge \text{linearform } H h$   
 $\wedge H \trianglelefteq E$   
 $\wedge F \trianglelefteq H$   
 $\wedge \text{graph } F f \subseteq \text{graph } H h$   
 $\wedge (\forall x \in H. h x \leq p x)$ }

**lemma** *norm-pres-extensionE* [elim]:

**assumes** g  $\in$  *norm-pres-extensions* E p F f

**obtains** H h

**where** g = *graph* H h

**and** *linearform* H h

**and**  $H \trianglelefteq E$

**and**  $F \trianglelefteq H$

**and** *graph* F f  $\subseteq$  *graph* H h

**and**  $\forall x \in H. h x \leq p x$

*<proof>*

**lemma** *norm-pres-extensionI2* [intro]:

*linearform* H h  $\Longrightarrow H \trianglelefteq E \Longrightarrow F \trianglelefteq H$   
 $\Longrightarrow \text{graph } F f \subseteq \text{graph } H h \Longrightarrow \forall x \in H. h x \leq p x$   
 $\Longrightarrow \text{graph } H h \in \text{norm-pres-extensions } E p F f$   
*<proof>*

**lemma** *norm-pres-extensionI*:

$\exists H h. g = \text{graph } H h$   
 $\wedge \text{linearform } H h$   
 $\wedge H \trianglelefteq E$   
 $\wedge F \trianglelefteq H$   
 $\wedge \text{graph } F f \subseteq \text{graph } H h$   
 $\wedge (\forall x \in H. h x \leq p x) \Longrightarrow g \in \text{norm-pres-extensions } E p F f$   
*<proof>*

**end**

## 8 The norm of a function

**theory** *Function-Norm*

**imports** *Normed-Space Function-Order*

**begin**

### 8.1 Continuous linear forms

A linear form  $f$  on a normed vector space  $(V, \|\cdot\|)$  is *continuous*, iff it is bounded, i.e.

$$\exists c \in R. \forall x \in V. |f x| \leq c \cdot \|x\|$$



In our application no other functions than linear forms are considered, so we can define continuous linear forms as bounded linear forms:

```
locale continuous = linearform +
  fixes norm :: - => real (||-||)
  assumes bounded:  $\exists c. \forall x \in V. |f x| \leq c * \|x\|$ 
```

```
declare continuous.intro [intro?] continuous-axioms.intro [intro?]
```

```
lemma continuousI [intro]:
  fixes norm :: - => real (||-||)
  assumes linearform V f
  assumes r:  $\bigwedge x. x \in V \implies |f x| \leq c * \|x\|$ 
  shows continuous V f norm
  <proof>
```

## 8.2 The norm of a linear form

The least real number  $c$  for which holds

$$\forall x \in V. |f x| \leq c \cdot \|x\|$$

is called the *norm* of  $f$ .

For non-trivial vector spaces  $V \neq \{0\}$  the norm can be defined as

$$\|f\| = \sup_{x \neq 0} |f x| / \|x\|$$

For the case  $V = \{0\}$  the supremum would be taken from an empty set. Since  $\mathbf{R}$  is unbounded, there would be no supremum. To avoid this situation it must be guaranteed that there is an element in this set. This element must be  $\{ \} \geq 0$  so that *fn-norm* has the norm properties. Furthermore it does not have to change the norm in all other cases, so it must be  $0$ , as all other elements are  $\{ \} \geq 0$ .

Thus we define the set  $B$  where the supremum is taken from as follows:

$$\{0\} \cup \{|f x| / \|x\|. x \neq 0 \wedge x \in F\}$$

*fn-norm* is equal to the supremum of  $B$ , if the supremum exists (otherwise it is undefined).

```
locale fn-norm =
  fixes norm :: - => real (||-||)
  fixes B defines B V f  $\equiv \{0\} \cup \{|f x| / \|x\| \mid x. x \neq 0 \wedge x \in V\}$ 
  fixes fn-norm (||-||-- [0, 1000] 999)
  defines ||f||-V  $\equiv \bigsqcup (B V f)$ 
```

```
locale normed-vectorspace-with-fn-norm = normed-vectorspace + fn-norm
```

```
lemma (in fn-norm) B-not-empty [intro]:  $0 \in B V f$ 
  <proof>
```

The following lemma states that every continuous linear form on a normed space  $(V, \|\cdot\|)$  has a function norm.

```
lemma (in normed-vectorspace-with-fn-norm) fn-norm-works:
```

**assumes** *continuous V f norm*  
**shows** *lub (B V f) (||f||-V)*  
*<proof>*

**lemma** (*in normed-vectorspace-with-fn-norm*) *fn-norm-ub [iff?]*:  
**assumes** *continuous V f norm*  
**assumes** *b: b ∈ B V f*  
**shows** *b ≤ ||f||-V*  
*<proof>*

**lemma** (*in normed-vectorspace-with-fn-norm*) *fn-norm-leastB*:  
**assumes** *continuous V f norm*  
**assumes** *b: ⋀ b. b ∈ B V f ⇒ b ≤ y*  
**shows** *||f||-V ≤ y*  
*<proof>*

The norm of a continuous function is always  $\geq 0$ .

**lemma** (*in normed-vectorspace-with-fn-norm*) *fn-norm-ge-zero [iff]*:  
**assumes** *continuous V f norm*  
**shows** *0 ≤ ||f||-V*  
*<proof>*

The fundamental property of function norms is:

$$|f x| \leq \|f\| \cdot \|x\|$$

**lemma** (*in normed-vectorspace-with-fn-norm*) *fn-norm-le-cong*:  
**assumes** *continuous V f norm linearform V f*  
**assumes** *x: x ∈ V*  
**shows** *|f x| ≤ ||f||-V \* ||x||*  
*<proof>*

The function norm is the least positive real number for which the following inequality holds:

$$|f x| \leq c \cdot \|x\|$$

**lemma** (*in normed-vectorspace-with-fn-norm*) *fn-norm-least [intro?]*:  
**assumes** *continuous V f norm*  
**assumes** *ineq: ⋀ x. x ∈ V ⇒ |f x| ≤ c \* ||x||* **and** *ge: 0 ≤ c*  
**shows** *||f||-V ≤ c*  
*<proof>*

**end**

## 9 Zorn's Lemma

**theory** *Zorn-Lemma*  
**imports** *Main*  
**begin**

Zorn's Lemmas states: if every linear ordered subset of an ordered set  $S$  has an upper bound in  $S$ , then there exists a maximal element in  $S$ . In our application,

$S$  is a set of sets ordered by set inclusion. Since the union of a chain of sets is an upper bound for all elements of the chain, the conditions of Zorn's lemma can be modified: if  $S$  is non-empty, it suffices to show that for every non-empty chain  $c$  in  $S$  the union of  $c$  also lies in  $S$ .

**theorem** *Zorn's-Lemma:*

**assumes**  $r: \bigwedge c. c \in \text{chains } S \implies \exists x. x \in c \implies \bigcup c \in S$

**and**  $aS: a \in S$

**shows**  $\exists y \in S. \forall z \in S. y \subseteq z \longrightarrow z = y$

*<proof>*

**end**

## Part II

# Lemmas for the Proof

## 10 The supremum wrt. the function order

**theory** *Hahn-Banach-Sup-Lemmas*  
**imports** *Function-Norm Zorn-Lemma*  
**begin**

This section contains some lemmas that will be used in the proof of the Hahn-Banach Theorem. In this section the following context is presumed. Let  $E$  be a real vector space with a seminorm  $p$  on  $E$ .  $F$  is a subspace of  $E$  and  $f$  a linear form on  $F$ . We consider a chain  $c$  of norm-preserving extensions of  $f$ , such that  $\bigcup c = \text{graph } H h$ . We will show some properties about the limit function  $h$ , i.e. the supremum of the chain  $c$ .

Let  $c$  be a chain of norm-preserving extensions of the function  $f$  and let  $\text{graph } H h$  be the supremum of  $c$ . Every element in  $H$  is member of one of the elements of the chain.

**lemmas** [*dest?*] = *chainsD*  
**lemmas** *chainsE2* [*elim?*] = *chainsD2* [*elim-format*]

**lemma** *some-H'h'*:

**assumes**  $M: M = \text{norm-pres-extensions } E p F f$   
**and**  $cM: c \in \text{chains } M$   
**and**  $u: \text{graph } H h = \bigcup c$   
**and**  $x: x \in H$   
**shows**  $\exists H' h'. \text{graph } H' h' \in c$   
 $\wedge (x, h x) \in \text{graph } H' h'$   
 $\wedge \text{linearform } H' h' \wedge H' \trianglelefteq E$   
 $\wedge F \trianglelefteq H' \wedge \text{graph } F f \subseteq \text{graph } H' h'$   
 $\wedge (\forall x \in H'. h' x \leq p x)$   
*<proof>*

Let  $c$  be a chain of norm-preserving extensions of the function  $f$  and let  $\text{graph } H h$  be the supremum of  $c$ . Every element in the domain  $H$  of the supremum function is member of the domain  $H'$  of some function  $h'$ , such that  $h$  extends  $h'$ .

**lemma** *some-H'h'*:

**assumes**  $M: M = \text{norm-pres-extensions } E p F f$   
**and**  $cM: c \in \text{chains } M$   
**and**  $u: \text{graph } H h = \bigcup c$   
**and**  $x: x \in H$   
**shows**  $\exists H' h'. x \in H' \wedge \text{graph } H' h' \subseteq \text{graph } H h$   
 $\wedge \text{linearform } H' h' \wedge H' \trianglelefteq E \wedge F \trianglelefteq H'$   
 $\wedge \text{graph } F f \subseteq \text{graph } H' h' \wedge (\forall x \in H'. h' x \leq p x)$   
*<proof>*

Any two elements  $x$  and  $y$  in the domain  $H$  of the supremum function  $h$  are both in the domain  $H'$  of some function  $h'$ , such that  $h$  extends  $h'$ .

**lemma** *some- $H'h'2$* :  
**assumes**  $M: M = \text{norm-pres-extensions } E \ p \ F \ f$   
**and**  $cM: c \in \text{chains } M$   
**and**  $u: \text{graph } H \ h = \bigcup c$   
**and**  $x: x \in H$   
**and**  $y: y \in H$   
**shows**  $\exists H' \ h'. x \in H' \wedge y \in H'$   
 $\wedge \text{graph } H' \ h' \subseteq \text{graph } H \ h$   
 $\wedge \text{linearform } H' \ h' \wedge H' \trianglelefteq E \wedge F \trianglelefteq H'$   
 $\wedge \text{graph } F \ f \subseteq \text{graph } H' \ h' \wedge (\forall x \in H'. h' \ x \leq p \ x)$   
 $\langle \text{proof} \rangle$

The relation induced by the graph of the supremum of a chain  $c$  is definite, i.e. it is the graph of a function.

**lemma** *sup-definite*:  
**assumes**  $M\text{-def}: M = \text{norm-pres-extensions } E \ p \ F \ f$   
**and**  $cM: c \in \text{chains } M$   
**and**  $xy: (x, y) \in \bigcup c$   
**and**  $xz: (x, z) \in \bigcup c$   
**shows**  $z = y$   
 $\langle \text{proof} \rangle$

The limit function  $h$  is linear. Every element  $x$  in the domain of  $h$  is in the domain of a function  $h'$  in the chain of norm preserving extensions. Furthermore,  $h$  is an extension of  $h'$  so the function values of  $x$  are identical for  $h'$  and  $h$ . Finally, the function  $h'$  is linear by construction of  $M$ .

**lemma** *sup-lf*:  
**assumes**  $M: M = \text{norm-pres-extensions } E \ p \ F \ f$   
**and**  $cM: c \in \text{chains } M$   
**and**  $u: \text{graph } H \ h = \bigcup c$   
**shows**  $\text{linearform } H \ h$   
 $\langle \text{proof} \rangle$

The limit of a non-empty chain of norm preserving extensions of  $f$  is an extension of  $f$ , since every element of the chain is an extension of  $f$  and the supremum is an extension for every element of the chain.

**lemma** *sup-ext*:  
**assumes**  $\text{graph}: \text{graph } H \ h = \bigcup c$   
**and**  $M: M = \text{norm-pres-extensions } E \ p \ F \ f$   
**and**  $cM: c \in \text{chains } M$   
**and**  $ex: \exists x. x \in c$   
**shows**  $\text{graph } F \ f \subseteq \text{graph } H \ h$   
 $\langle \text{proof} \rangle$

The domain  $H$  of the limit function is a superspace of  $F$ , since  $F$  is a subset of  $H$ . The existence of the  $0$  element in  $F$  and the closure properties follow from the fact that  $F$  is a vector space.

**lemma** *sup-supF*:  
**assumes**  $\text{graph}: \text{graph } H \ h = \bigcup c$   
**and**  $M: M = \text{norm-pres-extensions } E \ p \ F \ f$

```

and  $cM: c \in \text{chains } M$ 
and  $ex: \exists x. x \in c$ 
and  $FE: F \leq E$ 
shows  $F \leq H$ 
<proof>

```

The domain  $H$  of the limit function is a subspace of  $E$ .

```

lemma sup-subE:
assumes  $\text{graph}: \text{graph } H \ h = \bigcup c$ 
and  $M: M = \text{norm-pres-extensions } E \ p \ F \ f$ 
and  $cM: c \in \text{chains } M$ 
and  $ex: \exists x. x \in c$ 
and  $FE: F \leq E$ 
and  $E: \text{vector space } E$ 
shows  $H \leq E$ 
<proof>

```

The limit function is bounded by the norm  $p$  as well, since all elements in the chain are bounded by  $p$ .

```

lemma sup-norm-pres:
assumes  $\text{graph}: \text{graph } H \ h = \bigcup c$ 
and  $M: M = \text{norm-pres-extensions } E \ p \ F \ f$ 
and  $cM: c \in \text{chains } M$ 
shows  $\forall x \in H. h \ x \leq p \ x$ 
<proof>

```

The following lemma is a property of linear forms on real vector spaces. It will be used for the lemma *abs-Hahn-Banach* (see page 24). For real vector spaces the following inequality are equivalent:

$$\forall x \in H. |h \ x| \leq p \ x \quad \text{and} \quad \forall x \in H. h \ x \leq p \ x$$

```

lemma abs-ineq-iff:
assumes  $\text{subspace } H \ E$  and  $\text{vector space } E$  and  $\text{seminorm } E \ p$ 
and  $\text{linearform } H \ h$ 
shows  $(\forall x \in H. |h \ x| \leq p \ x) = (\forall x \in H. h \ x \leq p \ x)$  (is ?L = ?R)
<proof>

```

**end**

## 11 Extending non-maximal functions

```

theory Hahn-Banach-Ext-Lemmas
imports Function-Norm
begin

```

In this section the following context is presumed. Let  $E$  be a real vector space with a seminorm  $q$  on  $E$ .  $F$  is a subspace of  $E$  and  $f$  a linear function on  $F$ . We consider a subspace  $H$  of  $E$  that is a superspace of  $F$  and a linear form  $h$  on  $H$ .  $H$  is not equal to  $E$  and  $x_0$  is an element in  $E - H$ .  $H$  is extended to the direct sum  $H' = H + \text{lin } x_0$ , so for any  $x \in H'$  the decomposition of  $x = y +$

$a \cdot x$  with  $y \in H$  is unique.  $h'$  is defined on  $H'$  by  $h' x = h y + a \cdot \xi$  for a certain  $\xi$ .

Subsequently we show some properties of this extension  $h'$  of  $h$ .

This lemma will be used to show the existence of a linear extension of  $f$  (see page ??). It is a consequence of the completeness of  $\mathbf{R}$ . To show

$$\exists \xi. \forall y \in F. a y \leq \xi \wedge \xi \leq b y$$

it suffices to show that

$$\forall u \in F. \forall v \in F. a u \leq b v$$

**lemma** *ex-xi*:

**assumes** *vectorspace*  $F$

**assumes**  $r$ :  $\bigwedge u v. u \in F \implies v \in F \implies a u \leq b v$

**shows**  $\exists xi::real. \forall y \in F. a y \leq xi \wedge xi \leq b y$

*<proof>*

The function  $h'$  is defined as a  $h' x = h y + a \cdot \xi$  where  $x = y + a \cdot \xi$  is a linear extension of  $h$  to  $H'$ .

**lemma** *h'-lf*:

**assumes** *h'-def*:  $\bigwedge x. h' x = (let (y, a) =$

*SOME*  $(y, a). x = y + a \cdot x0 \wedge y \in H$  in  $h y + a * xi)$

**and** *H'-def*:  $H' = H + \text{lin } x0$

**and** *HE*:  $H \leq E$

**assumes** *linearform*  $H h$

**assumes**  $x0$ :  $x0 \notin H \ x0 \in E \ x0 \neq 0$

**assumes**  $E$ : *vectorspace*  $E$

**shows** *linearform*  $H' h'$

*<proof>*

The linear extension  $h'$  of  $h$  is bounded by the seminorm  $p$ .

**lemma** *h'-norm-pres*:

**assumes** *h'-def*:  $\bigwedge x. h' x = (let (y, a) =$

*SOME*  $(y, a). x = y + a \cdot x0 \wedge y \in H$  in  $h y + a * xi)$

**and** *H'-def*:  $H' = H + \text{lin } x0$

**and**  $x0$ :  $x0 \notin H \ x0 \in E \ x0 \neq 0$

**assumes**  $E$ : *vectorspace*  $E$  **and** *HE*: *subspace*  $H E$

**and** *seminorm*  $E p$  **and** *linearform*  $H h$

**assumes**  $a$ :  $\forall y \in H. h y \leq p y$

**and**  $a'$ :  $\forall y \in H. -p (y + x0) - h y \leq xi \wedge xi \leq p (y + x0) - h y$

**shows**  $\forall x \in H'. h' x \leq p x$

*<proof>*

**end**

## Part III

# The Main Proof

## 12 The Hahn-Banach Theorem

**theory** *Hahn-Banach*  
**imports** *Hahn-Banach-Lemmas*  
**begin**

We present the proof of two different versions of the Hahn-Banach Theorem, closely following [1, §36].

### 12.1 The Hahn-Banach Theorem for vector spaces

**Hahn-Banach Theorem.** Let  $F$  be a subspace of a real vector space  $E$ , let  $p$  be a semi-norm on  $E$ , and  $f$  be a linear form defined on  $F$  such that  $f$  is bounded by  $p$ , i.e.  $\forall x \in F. f x \leq p x$ . Then  $f$  can be extended to a linear form  $h$  on  $E$  such that  $h$  is norm-preserving, i.e.  $h$  is also bounded by  $p$ .

#### Proof Sketch.

1. Define  $M$  as the set of norm-preserving extensions of  $f$  to subspaces of  $E$ . The linear forms in  $M$  are ordered by domain extension.
2. We show that every non-empty chain in  $M$  has an upper bound in  $M$ .
3. With Zorn's Lemma we conclude that there is a maximal function  $g$  in  $M$ .
4. The domain  $H$  of  $g$  is the whole space  $E$ , as shown by classical contradiction:
  - Assuming  $g$  is not defined on whole  $E$ , it can still be extended in a norm-preserving way to a super-space  $H'$  of  $H$ .
  - Thus  $g$  can not be maximal. Contradiction!

**theorem** *Hahn-Banach*:

**assumes**  $E$ : *vectorspace*  $E$  **and** *subspace*  $F E$

**and** *seminorm*  $E p$  **and** *linearform*  $F f$

**assumes**  $fp$ :  $\forall x \in F. f x \leq p x$

**shows**  $\exists h. \text{linearform } E h \wedge (\forall x \in F. h x = f x) \wedge (\forall x \in E. h x \leq p x)$

— Let  $E$  be a vector space,  $F$  a subspace of  $E$ ,  $p$  a seminorm on  $E$ ,

— and  $f$  a linear form on  $F$  such that  $f$  is bounded by  $p$ ,

— then  $f$  can be extended to a linear form  $h$  on  $E$  in a norm-preserving way.

*<proof>*

### 12.2 Alternative formulation

The following alternative formulation of the Hahn-Banach Theorem uses the fact that for a real linear form  $f$  and a seminorm  $p$  the following inequality are equivalent:<sup>1</sup>

<sup>1</sup>This was shown in lemma *abs-ineq-iff* (see page 22).



$$\forall x \in H. |h x| \leq p x \quad \text{and} \quad \forall x \in H. h x \leq p x$$

**theorem** *abs-Hahn-Banach*:

**assumes** *E*: vectorspace *E* **and** *FE*: subspace *F E*

**and** *lf*: linearform *F f* **and** *sn*: seminorm *E p*

**assumes** *fp*:  $\forall x \in F. |f x| \leq p x$

**shows**  $\exists g$ . linearform *E g*

$\wedge (\forall x \in F. g x = f x)$

$\wedge (\forall x \in E. |g x| \leq p x)$

*<proof>*

## 12.3 The Hahn-Banach Theorem for normed spaces

Every continuous linear form *f* on a subspace *F* of a norm space *E*, can be extended to a continuous linear form *g* on *E* such that  $\|f\| = \|g\|$ .

**theorem** *norm-Hahn-Banach*:

**fixes** *V* **and** norm ( $\|\cdot\|$ )

**fixes** *B* **defines**  $\bigwedge V f. B V f \equiv \{0\} \cup \{|f x| / \|x\| \mid x. x \neq 0 \wedge x \in V\}$

**fixes** *fn-norm* ( $\|\cdot\|$ - [0, 1000] 999)

**defines**  $\bigwedge V f. \|f\|$ -*V*  $\equiv \bigsqcup (B V f)$

**assumes** *E-norm*: normed-vectorspace *E norm* **and** *FE*: subspace *F E*

**and** *linearform*: linearform *F f* **and** *continuous F f norm*

**shows**  $\exists g$ . linearform *E g*

$\wedge$  *continuous E g norm*

$\wedge (\forall x \in F. g x = f x)$

$\wedge \|g\|$ -*E* =  $\|f\|$ -*F*

*<proof>*

**end**

## References

- [1] H. Heuser. *Funktionalanalysis: Theorie und Anwendung*. Teubner, 1986.
- [2] L. Narici and E. Beckenstein. The Hahn-Banach Theorem: The life and times. In *Topology Atlas*. York University, Toronto, Ontario, Canada, 1996. <http://at.yorku.ca/topology/preprint.htm> and <http://at.yorku.ca/p/a/a/a/16.htm>.
- [3] B. Nowak and A. Trybulec. Hahn-Banach theorem. *Journal of Formalized Mathematics*, 5, 1993. <http://mizar.uwb.edu.pl/JFM/Vol5/hahmban.html>.